

CUDA raytracing algorithm for visualizing discrete element model output

Anders Damsgaard Christensen, 20062213

Abstract—A raytracing algorithm is constructed using the CUDA API for visualizing output from a CUDA discrete element model, which outputs spatial information in dynamic particle systems. The raytracing algorithm is optimized with constant memory and compilation flags, and performance is measured as a function of the number of particles and the number of pixels. The execution time is compared to equivalent CPU code, and the speedup under a variety of conditions is found to have a mean value of 55.6 times.

Index Terms—CUDA, discrete element method, raytracing

I. INTRODUCTION

VISUALIZING systems containing many spheres using traditional object-order graphics rendering can often result in very high computational requirements, as the usual automated approach is to construct a meshed surface with a specified resolution for each sphere. The memory requirements are thus quite high, as each surface will consist of many vertices. Raytracing [1] is a viable alternative, where spheric entities are saved as data structures with a centre coordinate and a radius. The rendering is performed on the base of these values, which results in a perfectly smooth surfaced sphere. To accelerate the rendering, the algorithm is constructed utilizing the CUDA API [2], where the problem is divided into $n \times m$ threads, corresponding to the desired output image resolution. Each thread iterates through all particles and applies a simple shading model to determine the final RGB values of the pixel.

Previous studies of GPU or CUDA implementations of ray tracing algorithms reported major speedups, compared to corresponding CPU applications (e.g. [3], [4], [5], [6]). None of the software was however found to be open-source and GPL licensed, so a simple raytracer was constructed, customized to render particles, where the data was stored in a specific data format.

A. Discrete Element Method

The input particle data to the raytracer is the output of a custom CUDA-based Discrete Element Method (DEM) application currently in development. The DEM model is used to numerically simulate the response of a drained, soft, granular sediment bed upon normal stresses and shearing velocities similar to subglacial environments under ice streams [7]. In contrast to laboratory experiments on granular material, the discrete element method [8] approach allows close monitoring of the progressive deformation, where all involved physical

parameters of the particles and spatial boundaries are readily available for continuous inspection.

The discrete element method (DEM) is a subtype of molecular dynamics (MD), and discretizes time into sufficiently small timesteps, and treats the granular material as discrete grains, interacting through contact forces. Between time steps, the particles are allowed to overlap slightly, and the magnitude of the overlap and the kinematic states of the particles is used to compute normal- and shear components of the contact force. The particles are treated as spherical entities, which simplifies the contact search. The spatial simulation domain is divided using a homogeneous, uniform, cubic grid, which greatly reduces the amount of possible contacts that are checked during each timestep. The grid-particle list is sorted using Thrust¹, and updated each timestep. The new particle positions and kinematic values are updated by inserting the resulting force and torque into Newton's second law, and using a Taylor-based second order integration scheme to calculate new linear and rotational accelerations, velocities and positions.

B. Application usage

The CUDA DEM application is a command line executable, and writes updated particle information to custom binary files with a specific interval. This raytracing algorithm is constructed to also run from the command line, be non-interactive, and write output images in the PPM image format. This format is chosen to allow rendering to take place on cluster nodes with CUDA compatible devices.

Both the CUDA DEM and raytracing applications are open-source², although still under heavy development.

This document consists of a short introduction to the basic mathematics behind the ray tracing algorithm, an explanation of the implementation using the CUDA API [2] and a presentation of the results. The CUDA device source code and C++ host source code for the ray tracing algorithm can be found in the appendix, along with instructions for compilation and execution of the application.

II. RAY TRACING ALGORITHM

The goal of the ray tracing algorithm is to compute the shading of each pixel in the image [9]. This is performed by creating a viewing ray from the eye into the scene, finding the closest intersection with a scene object, and computing the resulting color. The general structure of the program is demonstrated in the following pseudo-code:

Contact: anders.damsgaard@geo.au.dk

Webpage: <http://users-cs.au.dk/adc>

Manuscript, last revision: October 20, 2011.

¹<http://code.google.com/p/thrust/>

²<http://users-cs.au.dk/adc/files/sphere.tar.gz>

```

for each pixel do
  compute viewing ray origin and direction
  iterate through objects and find the closest hit
  set pixel color to value computed from hit  $\leftrightarrow$ 
  point, light, n

```

The implemented code does not utilize recursive rays, since the modeled material grains are matte in appearance.

A. Ray generation

The rays are in vector form defined as:

$$\mathbf{p}(t) = \mathbf{e} + t(\mathbf{s} - \mathbf{e}) \quad (1)$$

The perspective can be either *orthographic*, where all viewing rays have the same direction, but different starting points, or use *perspective projection*, where the starting point is the same, but the direction is slightly different [9]. For the purposes of this application, a perspective projection was chosen, as it results in the most natural looking image. The ray data structures were held flexible enough to allow an easy implementation of orthographic perspective, if this is desired at a later point.

The ray origin \mathbf{e} is the position of the eye, and is constant. The direction is unique for each ray, and is computed using:

$$\mathbf{s} - \mathbf{e} = -d\mathbf{w} + u\mathbf{u} + v\mathbf{v} \quad (2)$$

where $\{\mathbf{u}, \mathbf{v}, \mathbf{w}\}$ are the orthonormal bases of the camera coordinate system, and d is the focal length [9]. The camera coordinates of pixel (i, j) in the image plane, u and v , are calculated by:

$$u = l + (r - l)(i + 0.5)/n$$

$$v = b + (t - b)(j + 0.5)/m$$

where l, r, t and b are the positions of the image borders (left, right, top and bottom) in camera space. The values n and m are the number of pixels in each dimension.

B. Ray-sphere intersection

Given a sphere with a center \mathbf{c} , and radius R , a equation can be constrained, where \mathbf{p} are all points placed on the sphere surface:

$$(\mathbf{p} - \mathbf{c}) \cdot (\mathbf{p} - \mathbf{c}) - R^2 = 0 \quad (3)$$

By substituting the points \mathbf{p} with ray equation 1, and rearranging the terms, a quadratic equation emerges:

$$(\mathbf{d} \cdot \mathbf{d})t^2 + 2\mathbf{d} \cdot (\mathbf{e} - \mathbf{c})t + (\mathbf{e} - \mathbf{c}) \cdot (\mathbf{e} - \mathbf{c}) - R^2 = 0 \quad (4)$$

The number of ray steps t is the only unknown, so the number of intersections is found by calculating the determinant:

$$\Delta = (2(\mathbf{d} \cdot (\mathbf{e} - \mathbf{c})))^2 - 4(\mathbf{d} \cdot \mathbf{d})((\mathbf{e} - \mathbf{c}) \cdot (\mathbf{e} - \mathbf{c}) - R^2) \quad (5)$$

A negative value denotes no intersection between the sphere and the ray, a value of zero means that the ray touches the sphere at a single point (ignored in this implementation), and a positive value denotes that there are two intersections, one when the ray enters the sphere, and one when it exits. In the code, a conditional branch checks whether the determinant is

positive. If this is the case, the distance to the intersection in ray “steps” is calculated using:

$$t = \frac{-\mathbf{d} \cdot (\mathbf{e} - \mathbf{c}) \pm \sqrt{\eta}}{(\mathbf{d} \cdot \mathbf{d})} \quad (6)$$

where

$$\eta = (\mathbf{d} \cdot (\mathbf{e} - \mathbf{c}))^2 - (\mathbf{d} \cdot \mathbf{d})((\mathbf{e} - \mathbf{c}) \cdot (\mathbf{e} - \mathbf{c}) - R^2)$$

Only the smallest intersection (t_{minus}) is calculated, since this marks the point where the sphere enters the particle. If this value is smaller than previous intersection distances, the intersection point \mathbf{p} and surface normal \mathbf{n} at the intersection point is calculated:

$$\mathbf{p} = \mathbf{e} + t_{\text{minus}}\mathbf{d} \quad (7)$$

$$\mathbf{n} = 2(\mathbf{p} - \mathbf{c}) \quad (8)$$

The intersection distance in vector steps (t_{minus}) is saved in order to allow comparison of the distance with later intersections.

C. Pixel shading

The pixel is shaded using *Lambertian* shading [9], where the pixel color is proportional to the angle between the light vector (l) and the surface normal. An ambient shading component is added to simulate global illumination, and prevent that the spheres are completely black:

$$L = k_a I_a + k_d I_d \max(0, (\mathbf{n} \cdot \mathbf{l})) \quad (9)$$

where the a and d subscripts denote the ambient and diffusive (Lambertian) components of the ambient/diffusive coefficients (k) and light intensities (I). The pixel color L is calculated once per color channel.

D. Computational implementation

The above routines were first implemented in CUDA for device execution, and afterwards ported to a CPU C++ equivalent, used for comparing performance. The CPU raytracing algorithm was optimized to shared-memory parallelism using OpenMP [10]. The execution method can be chosen when launching the raytracer from the command line, see the appendix for details. In the CPU implementation, all data was stored in linear arrays of the right size, ensuring 100% memory efficiency.

III. CUDA IMPLEMENTATION

When constructing the algorithm for execution on the GPGPU device, the data-parallel nature of the problem (SIMD: single instruction, multiple data) is used to deconstruct the rendering task into a single thread per pixel. Each thread iterates through all particles, and ends up writing the resulting color to the image memory.

The application starts by reading the discrete element method data from a custom binary file. The particle data, consisting of position vectors in three-dimensional Euclidean space (\mathbf{R}^3) and particle radii, is stored together in a `float4` array, with the particle radius in the `w` position. This has




Fig. 1. Sample output of GPU raytracer rendering of 512 particles.




Fig. 2. Sample output of GPU raytracer rendering of 50 653 particles.

large advantages to storing the data in separate `float3` and `float` arrays; Using `float4` (instead of `float3`) data allows coalesced memory access [11] to the arrays of data in device memory, resulting in efficient memory requests and transfers [12], and the data access pattern is coherent and convenient. Other three-component vectors were also stored as `float4` for the same reasons, even though this sometimes caused a slight memory redundancy. The image data is saved in a three-channel linear unsigned `char` array. Global memory access are coalesced whenever possible. Divergent branches in the kernel code were avoided as much as possible [11].

The algorithm starts by allocating memory on the device for the particle data, the ray parameters, and the image RGB values. Afterwards, all particle data is transferred from the host- to the device memory.

All pixel values are initialized to $[R, G, B] = [0, 0, 0]$, which serves as the image background color. Afterwards, a kernel is executed with a thread for each pixel, testing for intersections between the pixel's viewing ray and all particles, and returning the closest particle. This information is used when computing the shading of the pixel.

After all pixel values have been computed, the image data is transferred back to the host memory, and written to the disk. The application ends by liberating dynamically allocated memory on both the device and the host.

A. Thread and block layout

The thread/block layout passed during kernel launches is arranged in the following manner:

```
dim3 threads(16, 16);
dim3 blocks((width+15)/16, (height+15)/16);
```

The image pixel position of the thread can be determined from the thread- and block index and dimensions. The layout corresponds to a thread tile size of 256, and a dynamic number of blocks, ensured to fit the image dimensions with only small eventual redundancy [13]. Since this method will initialize extra threads in most situations, all kernels (with return type `void`) start by checking whether the thread-/block index actually falls inside of the image dimensions:

```
int i = threadIdx.x + blockIdx.x * ←
    blockDim.x;
int j = threadIdx.y + blockIdx.y * ←
    blockDim.y;
unsigned int mempos = x + y * blockDim.x ←
    * gridDim.x;
if (mempos > pixels)
    return;
```

The linear memory position (`mempos`) is used as the index when reading or writing to the linear arrays residing in global device memory.

B. Image output

After completing all pixel shading computations on the device, the image data is transferred back to the host memory, and together with a header written to a PPM³ image file. This file is converted to the PNG format using ImageMagick.

C. Performance

Since this simple raytracing algorithm generates a single non-recursive ray for each pixel, which in turn checks all spheres for intersection, the application is expected to scale in the form of $O(n \times m \times N)$, where n and m are the output image dimensions in pixels, and N is the number of particles.

³<http://paulbourke.net/dataformats/ppm/>




Fig. 3. Performance scaling with varying particle numbers at image dimensions 800 by 800 pixels.




Fig. 4. Performance scaling with varying image dimensions ($n \times m$) with 5832 particles.

The data transfer between the host and device is kept at a bare minimum, as the intercommunication is considered a bottleneck in relation to the potential device performance[11]. Thread synchronization points are only inserted where necessary, and the code is optimized by the compilers to the target architecture (see appendix).

The host execution time was profiled using a `clock()` based CPU timer from `time.h`, which was normalized using the constant `CLOCKS_PER_SEC`.

The device execution time was profiled using two `cudaEvent_t` timers, one measuring the time spent in the entire device code section, including device memory allocation, data transfer to- and from the host, execution of the kernels, and memory deallocation. The other timer only measured time spent in the kernels. The threads were synchronized before stopping the timers. A simple CPU timer using `clock()` will *not* work, since control is returned to the host thread before the device code has completed all tasks.

Figures 3 and 4 show the profiling results, where the number of particles and the image dimensions were varied. With exception of executions with small image dimensions, the kernel execution time results agree with the $O(n \times m \times N)$ scaling prediction.

The device memory allocation and data transfer was also profiled, and turns out to be only weakly dependant on the particle numbers (fig. 3), but more strongly correlated to image dimensions (fig. 4). As with kernel execution times, the execution time converges against an overhead value at small image dimensions.

The CPU time spent in the host kernels proves to be linear with the particle numbers, and linear with the image dimensions. This is due to the non-existent overhead caused by initialization of the device code, and reduced memory transfer.

The ratio between CPU computational times and the sum of the device kernel execution time and the host—device memory transfer and additional memory allocation was calculated, and had a mean value of 55.6 and a variance of 739 out of the 11 comparative measurements presented in the figures. It should be noted, that the smallest speedups were recorded when using very small image dimensions, probably unrealistic in real use.

As the number of particles are not known by compilation, it is not possible to store particle positions and -radii in constant memory. Shared memory was also on purpose avoided, since the memory per thread block (64 kb) would not be sufficient in rendering of simulations containing more than 16 000 particles (16 000 `float4` values). The constant memory was however utilized for storing the camera related parameters; the orthonormal base vectors, the observer position, the image dimensions, the focal length, and the light vector.

Previous GPU implementations often rely on k-D trees, constructed as an sorting method for static scene objects[3], [5]. A k-D tree implementation would drastically reduce the global memory access induced by each thread, so it is therefore the next logical step with regards to optimizing the ray tracing algorithm presented here.

IV. CONCLUSION

This document presented the implementation of a basic ray tracing algorithm, utilizing the highly data-parallel nature of the problem when porting the work load to CUDA. Performance tests showed the expected, linear correlation between image dimensions, particle numbers and execution time. Comparisons with an equivalent CPU algorithm showed large speedups, typically up to two orders of magnitude. This speedup did not come at a cost of less correct results.

The final product will come into good use during further development and completion of the CUDA DEM particle model, and is ideal since it can be used for offline rendering on dedicated, heterogeneous GPU-CPU computing nodes. The included device code will be the preferred method of execution, whenever the host system allows it.

REFERENCES

- [1] T. Whitted, "An improved illumination model for shaded display," *Communications of the ACM*, vol. 23, no. 6, pp. 343–349, 1980.
- [2] NVIDIA, *CUDA C Programming Guide*, 3rd ed., NVIDIA Corporation: Santa Clara, CA, USA, Oct 2010.
- [3] D. Horn, J. Sugerman, M. Houston, and P. Hanrahan, "Interactive k-D tree GPU raytracing," *Association for Computing Machinery, Inc.*, pp. 167–174 pp., 2007.

- [4] M. Shih, Y. Chiu, Y. Chen, and C. Chang, "Real-time ray tracing with cuda," *Algorithms and Architectures for Parallel Processing*, pp. 327–337, 2009.
- [5] S. Popov, J. Günther, H. Seidel, and P. Slusallek, "Stackless kd-tree traversal for high performance gpu ray tracing," in *Computer Graphics Forum*, vol. 26, no. 3. Wiley Online Library, 2007, pp. 415–424.
- [6] D. Luebke and S. Parker, "Interactive ray tracing with cuda," *NVIDIA Technical Presentation, SIGGRAPH*, 2008.
- [7] D. Evans, E. Phillips, J. Hiemstra, and C. Auton, "Subglacial till: formation, sedimentary characteristics and classification," *Earth-Science Reviews*, vol. 78, no. 1-2, pp. 115–176, 2006.
- [8] P. Cundall and O. Strack, "A discrete numerical model for granular assemblies," *Géotechnique*, vol. 29, pp. 47–65, 1979.
- [9] P. Shirley, M. Ashikhmin, M. Gleicher, S. Marschner, E. Reinhard, K. Sung, W. Thompson, and P. Willemsen, *Fundamentals of computer graphics*, 3rd ed. AK Peters, 2009.
- [10] B. Chapman, G. Jost, and R. Van Der Pas, *Using OpenMP: portable shared memory parallel programming*. The MIT Press, 2007, vol. 10.
- [11] NVIDIA, *CUDA C Best Practices Guide Version*, 3rd ed., NVIDIA Corporation: Santa Clara, CA, USA, Aug 2010, cA Patent 95,050.
- [12] L. Nyland, M. Harris, and J. Prins, "Fast n-body simulation with cuda," *GPU gems*, vol. 3, pp. 677–695, 2007.
- [13] J. Sanders and E. Kandrot, *CUDA by example*. Addison-Wesley, 2010.

APPENDIX A TEST ENVIRONMENT

The raytracing algorithm was developed, tested and profiled on a mid 2010 Mac Pro with a 2.8 Ghz Quad-Core Intel Xeon CPU and a NVIDIA Quadro 4000 for Mac, dedicated to CUDA applications. The CUDA driver was version 4.0.50, the CUDA compilation tools release 4.0, V0.2.1221. The GCC tools were version 4.2.1. Each CPU core is multithreaded by two threads for a total of 8 threads.

The CUDA source code was compiled with `nvcc`, and linked to `g++` compiled C++ source code with `g++`. For all benchmark tests, the code was compiled with the following commands:

```
g++ -c -Wall -O3 -arch x86_64 -fopenmp ...
nvcc -c -use_fast_math -gencode <
    arch=compute_20,code=sm_20 -m64 ...
g++ -arch x86_64 -lcuda -lcudart -fopenmp <
    *.o -o rt
```

When profiling device code performance, the application was executed two times, and the time of the second run was noted. This was performed to avoid latency caused by device driver initialization.

The host system was measured to have a memory bandwidth of 4642.1 MB/s when transferring data from the host to the device, and 3805.6 MB/s when transferring data from the device to the host.

APPENDIX B SOURCE CODE

The entire source code, as well as input data files, can be found in the following archive <http://users-cs.au.dk/adc/files/sphere-rt.tar.gz>. The source code is built and run with the commands:

```
make
make run
```

With the `make run` command, the Makefile uses ImageMagick to convert the PPM file to PNG format, and the `OS X` command `open` to display the image. Other input data files are

included with other particle number magnitudes. The syntax for the raytracer application is the following:

```
./rt <CPU | GPU> <sphere-binary.bin> <
    <width> <height> <output-image.ppm>
```

This appendix contains the following source code files:

LISTINGS

1	rt_kernel.h	5
2	rt_kernel.cu	5
3	rt_kernel_cpu.h	7
4	rt_kernel_cpu.cpp	7

A. CUDA raytracing source code

Listing 1. rt_kernel.h

```
1 #ifndef RT_KERNEL_H_
2 #define RT_KERNEL_H_
3
4 #include <vector_functions.h>
5
6 // Constants
7 __constant__ float4 const_u;
8 __constant__ float4 const_v;
9 __constant__ float4 const_w;
10 __constant__ float4 const_eye;
11 __constant__ float4 const_imgplane;
12 __constant__ float const_d;
13 __constant__ float4 const_light;
14
15 // Host prototype functions
16
17 extern "C"
18 void cameraInit(float4 eye, float4 lookat, float <
    imgw, float hw_ratio);
19
20
21 extern "C"
22 void checkForCudaErrors(const char* <
    checkpoint_description);
23
24 extern "C"
25 int rt(float4* p, const unsigned int np,
26     rgb* img, const unsigned int width, const <
    unsigned int height,
27     f3 origo, f3 L, f3 eye, f3 lookat, float imgw);
28
29 #endif
```

Listing 2. rt_kernel.cu

```
1 #include <iostream>
2 #include <util_math.h>
3 #include "header.h"
4 #include "rt_kernel.h"
5
6 __inline__ __host__ __device__ float3 f4_to_f3(float4 <
    in)
7 {
8     return make_float3(in.x, in.y, in.z);
9 }
10
11 __inline__ __host__ __device__ float4 f3_to_f4(float3 <
    in)
12 {
13     return make_float4(in.x, in.y, in.z, 0.0f);
14 }
15
16 // Kernel for initializing image data
17 __global__ void imageInit(unsigned char* _img, <
    unsigned int pixels)
18 {
19     // Compute pixel position from threadIdx/blockIdx
20     int x = threadIdx.x + blockIdx.x * blockDim.x;
21     int y = threadIdx.y + blockIdx.y * blockDim.y;
22     unsigned int mempos = x + y * blockDim.x * gridDim.x;
23     if (mempos > pixels)
24         return;
```



```

25
26 _img[mempos*4] = 0; // Red channel
27 _img[mempos*4 + 1] = 0; // Green channel
28 _img[mempos*4 + 2] = 0; // Blue channel
29 }
30
31 // Calculate ray origins and directions
32 __global__ void rayInitPerspective(float4* _ray_origo,
33 float4* _ray_direction,
34 float4 eye,
35 unsigned int width,
36 unsigned int height)
37 {
38 // Compute pixel position from threadIdx/blockIdx
39 int i = threadIdx.x + blockIdx.x * blockDim.x;
40 int j = threadIdx.y + blockIdx.y * blockDim.y;
41 unsigned int mempos = i + j * blockDim.x * gridDim.x;
42 if (mempos > width*height)
43 return;
44
45 // Calculate pixel coordinates in image plane
46 float p_u = const_imgplane.x + (const_imgplane.y - ←
const_imgplane.x)
47 * (i + 0.5f) / width;
48 float p_v = const_imgplane.z + (const_imgplane.w - ←
const_imgplane.z)
49 * (j + 0.5f) / height;
50
51 // Write ray origo and direction to global memory
52 _ray_origo[mempos] = const_eye;
53 _ray_direction[mempos] = -const_d*const_w + ←
p_u*const_u + p_v*const_v;
54 }
55
56 // Check whether the pixel's viewing ray intersects ←
with the spheres,
57 // and shade the pixel correspondingly
58 __global__ void rayIntersectSpheres(float4* _ray_origo,
59 float4* ←
_ray_direction,
60 float4* _p,
61 unsigned char* _img,
62 unsigned int pixels,
63 unsigned int np)
64 {
65 // Compute pixel position from threadIdx/blockIdx
66 int x = threadIdx.x + blockIdx.x * blockDim.x;
67 int y = threadIdx.y + blockIdx.y * blockDim.y;
68 unsigned int mempos = x + y * blockDim.x * gridDim.x;
69 if (mempos > pixels)
70 return;
71
72 // Read ray data from global memory
73 float3 e = f4_to_f3(_ray_origo[mempos]);
74 float3 d = f4_to_f3(_ray_direction[mempos]);
75 // float step = length(d);
76
77 // Distance, in ray steps, between object and eye ←
initialized with a large value
78 float tdist = 1e10f;
79
80 // Surface normal at closest sphere intersection
81 float3 n;
82
83 // Intersection point coordinates
84 float3 p;
85
86 // Iterate through all particles
87 for (unsigned int i=0; i<np; i++) {
88 // Read sphere coordinate and radius
89 float3 c = f4_to_f3(_p[i]);
90 float R = _p[i].w;
91
92 // Calculate the discriminant: d = B^2 - 4AC
93 float Delta = ←
(2.0f*dot(d,(e-c)))*(2.0f*dot(d,(e-c))) // B^2
94 - 4.0f*dot(d,d) // -4*A
95 * (dot((e-c),(e-c)) - R*R); // C
96
97 // If the determinant is positive, there are two ←
solutions
98 // One where the line enters the sphere, and one ←
where it exits
99 if (Delta > 0.0f) {
100
101 // Calculate roots, Shirley 2009 p. 77
102 float t_minus = ((dot(-d,(e-c)) - sqrt( ←
dot(d,(e-c))*dot(d,(e-c)) - dot(d,d)
103 * (dot((e-c),(e-c)) - R*R) ) ) / ←
dot(d,d));
104
105 // Check whether intersection is closer than ←
previous values
106 if (fabs(t_minus) < tdist) {
107 p = e + t_minus*d;
108 tdist = fabs(t_minus);
109 n = normalize(2.0f * (p - c)); // Surface normal
110 }
111 } // End of solution branch
112 } // End of particle loop
113
114 // Write pixel color
115 if (tdist < 1e10) {
116 // Lambertian shading parameters
117 float dotprod = fabs(dot(n, f4_to_f3(const_light)));
118 float I_d = 40.0f; // Light intensity
119 float k_d = 5.0f; // Diffuse coefficient
120
121 // Ambient shading
122 float k_a = 10.0f;
123 float I_a = 5.0f;
124
125 // Write shading model values to pixel color ←
channels
126 _img[mempos*4] = (unsigned char) ((k_d * I_d ←
* dotprod
127 + k_a * I_a)*0.48f);
128 _img[mempos*4 + 1] = (unsigned char) ((k_d * I_d ←
* dotprod
129 + k_a * I_a)*0.41f);
130 _img[mempos*4 + 2] = (unsigned char) ((k_d * I_d ←
* dotprod
131 + k_a * I_a)*0.27f);
132 }
133 }
134
135 extern "C"
136 __host__ void cameraInit(float4 eye, float4 lookat, ←
float imgw, float hw_ratio)
137 {
138 // Image dimensions in world space (l, r, b, t)
139 float4 imgplane = make_float4(-0.5f*imgw, ←
0.5f*imgw, -0.5f*imgw*hw_ratio, ←
0.5f*imgw*hw_ratio);
140
141 // The view vector
142 float4 view = eye - lookat;
143
144 // Construct the camera view orthonormal base
145 float4 v = make_float4(0.0f, 1.0f, 0.0f, 0.0f); // ←
v: Pointing upward
146 float4 w = -view/length(view); // w: ←
Pointing backwards
147 float4 u = make_float4(cross(make_float3(v.x, v.y, ←
v.z),
148 make_float3(w.x, w.y, w.z)),
149 0.0f); // u: Pointing right
150
151 // Focal length 20% of eye vector length
152 float d = length(view)*0.8f;
153
154 // Light direction (points towards light source)
155 float4 light = ←
normalize(-1.0f*eye*make_float4(1.0f, 0.2f, ←
0.6f, 0.0f));
156
157 std::cout << "Transferring camera values to ←
constant memory\n";
158
159 cudaMemcpyToSymbol("const_u", &u, sizeof(u));
160 cudaMemcpyToSymbol("const_v", &v, sizeof(v));
161 cudaMemcpyToSymbol("const_w", &w, sizeof(w));
162 cudaMemcpyToSymbol("const_eye", &eye, sizeof(eye));
163 cudaMemcpyToSymbol("const_imgplane", &imgplane, ←
sizeof(imgplane));
164 cudaMemcpyToSymbol("const_d", &d, sizeof(d));

```

```

170     cudaMemcpyToSymbol("const_light", &light, ←
        sizeof(light));
171 }
172 // Check for CUDA errors
173 extern "C"
174 __host__ void checkForCudaErrors(const char* ←
        checkpoint_description)
175 {
176     cudaError_t err = cudaGetLastError();
177     if (err != cudaSuccess) {
178         std::cout << "\nCUDA_error_detected, ←
            checkpoint: " << ←
            " << checkpoint_description
179         << "\nError_string: " << ←
            cudaGetErrorString(err) << "\n";
180         exit(EXIT_FAILURE);
181     }
182 }
183 }
184 // Wrapper for the rt kernel
185 extern "C"
186 __host__ int rt(float4* p, unsigned int np,
187                rgb* img, unsigned int width, ←
                unsigned int height,
188                f3 origo, f3 L, f3 eye, f3 lookat, float ←
                imgw) {
189
190     using std::cout;
191
192     cout << "Initializing CUDA:\n";
193
194     // Initialize GPU timestamp recorders
195     float t1, t2;
196     cudaEvent_t t1_go, t2_go, t1_stop, t2_stop;
197     cudaEventCreate(&t1_go);
198     cudaEventCreate(&t2_go);
199     cudaEventCreate(&t1_stop);
200     cudaEventCreate(&t2_stop);
201     cudaEventCreate(&t1_stop);
202
203     // Start timer 1
204     cudaEventRecord(t1_go, 0);
205
206     // Allocate memory
207     cout << "Allocating device memory\n";
208     static float4 *p; // Particle positions ←
209     (x,y,z) and radius (w)
210     static unsigned char *img; // RGBw values in ←
211     image
212     static float4 *ray_origo; // Ray origo (x,y,z)
213     static float4 *ray_direction; // Ray direction ←
214     (x,y,z)
215     cudaMalloc((void**)&p, np*sizeof(float4));
216     cudaMalloc((void**)&img, ←
217         width*height*4*sizeof(unsigned char));
218     cudaMalloc((void**)&ray_origo, ←
219         width*height*sizeof(float4));
220     cudaMalloc((void**)&ray_direction, ←
221         width*height*sizeof(float4));
222
223     // Transfer particle data
224     cout << "Transferring particle_data: ←
225     device\n";
226     cudaMemcpy(_p, p, np*sizeof(float4), ←
227         cudaMemcpyHostToDevice);
228
229     // Check for errors after memory allocation
230     checkForCudaErrors("CUDA_error_after_memory ←
231     allocation");
232
233     // Arrange thread/block structure
234     unsigned int pixels = width*height;
235     float hw_ratio = (float)height/(float)width;
236     dim3 threads(16,16);
237     dim3 blocks((width+15)/16, (height+15)/16);
238
239     // Start timer 2
240     cudaEventRecord(t2_go, 0);
241
242     // Initialize image to background color
243     imageInit<<< blocks, threads >>>(_img, pixels);
244
245     // Initialize camera
246     cameraInit(make_float4(eye.x, eye.y, eye.z, 0.0f),
247                make_float4(lookat.x, lookat.y, ←
248                lookat.z, 0.0f),

```

```

240     imgw, hw_ratio);
241     checkForCudaErrors("CUDA_error_after_cameraInit");
242
243     // Construct rays for perspective projection
244     rayInitPerspective<<< blocks, threads >>>(
245         _ray_origo, _ray_direction,
246         make_float4(eye.x, eye.y, eye.z, 0.0f),
247         width, height);
248
249     // Find closest intersection between rays and spheres
250     rayIntersectSpheres<<< blocks, threads >>>(
251         _ray_origo, _ray_direction,
252         _p, _img, pixels, np);
253
254     // Make sure all threads are done before continuing ←
255     CPU control sequence
256     cudaThreadSynchronize();
257
258     // Check for errors
259     checkForCudaErrors("CUDA_error_after_kernel ←
260     execution");
261
262     // Stop timer 2
263     cudaEventRecord(t2_stop, 0);
264     cudaEventSynchronize(t2_stop);
265
266     // Transfer image data from device to host
267     cout << "Transferring image_data: ←
268     device → host\n";
269     cudaMemcpy(img, _img, ←
270         width*height*4*sizeof(unsigned char), ←
271         cudaMemcpyDeviceToHost);
272
273     // Free dynamically allocated device memory
274     cudaFree(_p);
275     cudaFree(_img);
276     cudaFree(_ray_origo);
277     cudaFree(_ray_direction);
278
279     // Stop timer 1
280     cudaEventRecord(t1_stop, 0);
281     cudaEventSynchronize(t1_stop);
282
283     // Calculate time spent in t1 and t2
284     cudaEventElapsedTime(&t1, t1_go, t1_stop);
285     cudaEventElapsedTime(&t2, t2_go, t2_stop);
286
287     // Report time spent
288     cout << "Time spent on entire GPU routine: "
289     << t1 << "ms\n";
290     cout << "Kernels: " << t2 << "ms\n"
291     << "Memory alloc. and transfer: " << t1-t2 ←
292     << "ms\n";
293
294     // Return successfully
295     return 0;
296 }

```

B. CPU raytracing source code

Listing 3. rt_kernel_cpu.h

```

1 #ifndef RT_KERNEL_CPU_H
2 #define RT_KERNEL_CPU_H
3
4 #include <vector_functions.h>
5
6 // Host prototype functions
7
8 void cameraInit(float3 eye, float3 lookat, float ←
9     imgw, float hw_ratio);
10
11 int rt_cpu(float4* p, const unsigned int np,
12            rgb* img, const unsigned int width, const ←
13            unsigned int height,
14            f3 origo, f3 L, f3 eye, f3 lookat, float imgw);
15 #endif

```

Listing 4. rt_kernel_cpu.cpp

```

1 #include <iostream>
2 #include <cstdlib>
3 #include <cmath>
4 #include <time.h>
5 #include <cuda.h>

```

```

6  #include <cutil_math.h>
7  #include <string.h>
8  #include "header.h"
9  #include "rt_kernel_cpu.h"
10
11 // Constants
12 float3 constc_u;
13 float3 constc_v;
14 float3 constc_w;
15 float3 constc_eye;
16 float4 constc_imgplane;
17 float constc_d;
18 float3 constc_light;
19
20 __inline__ float3 f4_to_f3(float4 in)
21 {
22     return make_float3(in.x, in.y, in.z);
23 }
24
25 __inline__ float4 f3_to_f4(float3 in)
26 {
27     return make_float4(in.x, in.y, in.z, 0.0f);
28 }
29
30 __inline__ float lengthf3(float3 in)
31 {
32     return sqrt(in.x*in.x + in.y*in.y + in.z*in.z);
33 }
34
35 // Kernel for initializing image data
36 void imageInit_cpu(unsigned char* _img, unsigned int ←
    pixels)
37 {
38     for (unsigned int mempos=0; mempos<pixels; ←
        mempos++) {
39         _img[mempos*4] = 0; // Red channel
40         _img[mempos*4 + 1] = 0; // Green channel
41         _img[mempos*4 + 2] = 0; // Blue channel
42     }
43 }
44
45 // Calculate ray origins and directions
46 void rayInitPerspective_cpu(float3* _ray_origo,
47     float3* _ray_direction,
48     float3 eye,
49     unsigned int width,
50     unsigned int height)
51 {
52     int i;
53     #pragma omp parallel for
54     for (i=0; i<width; i++) {
55         for (unsigned int j=0; j<height; j++) {
56             unsigned int mempos = i + j*height;
57
58             // Calculate pixel coordinates in image plane
59             float p_u = constc_imgplane.x + ←
                (constc_imgplane.y - constc_imgplane.x)
60             * (i + 0.5f) / width;
61             float p_v = constc_imgplane.z + ←
                (constc_imgplane.w - constc_imgplane.z)
62             * (j + 0.5f) / height;
63
64             // Write ray origo and direction to global memory
65             _ray_origo[mempos] = constc_eye;
66             _ray_direction[mempos] = -constc_d*constc_w + ←
                p_u*constc_u + p_v*constc_v;
67         }
68     }
69 }
70
71 // Check wether the pixel's viewing ray intersects ←
    with the spheres,
72 // and shade the pixel correspondingly
73 void rayIntersectSpheres_cpu(float3* _ray_origo,
74     float3* _ray_direction,
75     float4* _p,
76     unsigned char* _img,
77     unsigned int pixels,
78     unsigned int np)
79 {
80     int mempos;
81     #pragma omp parallel for
82     for (mempos=0; mempos<pixels; mempos++) {
83         // Read ray data from global memory
84
85         float3 e = _ray_origo[mempos];
86         float3 d = _ray_direction[mempos];
87         //float step = lengthf3(d);
88
89         // Distance, in ray steps, between object and eye ←
            initialized with a large value
90         float tdist = 1e10f;
91
92         // Surface normal at closest sphere intersection
93         float3 n;
94
95         // Intersection point coordinates
96         float3 p;
97
98         // Iterate through all particles
99         for (unsigned int i=0; i<np; i++) {
100             // Read sphere coordinate and radius
101             float3 c = f4_to_f3(_p[i]);
102             float R = _p[i].w;
103
104             // Calculate the discriminant: d = B^2 - 4AC
105             float Delta = ←
                (2.0f*dot(d,(e-c)))*(2.0f*dot(d,(e-c))) ←
                // B^2
106             - 4.0f*dot(d,d) // -4*A
107             * (dot((e-c),(e-c)) - R*R); // C
108
109             // If the determinant is positive, there are ←
                two solutions
110             // One where the line enters the sphere, and ←
                one where it exits
111             if (Delta > 0.0f) {
112                 // Calculate roots, Shirley 2009 p. 77
113                 float t_minus = ((dot(-d,(e-c)) - sqrt( ←
                    dot(d,(e-c))*dot(d,(e-c)) - dot(d,d)
114                 * (dot((e-c),(e-c)) - R*R) ) ) / dot(d,d));
115
116                 // Check wether intersection is closer than ←
                    previous values
117                 if (fabs(t_minus) < tdist) {
118                     p = e + t_minus*d;
119                     tdist = fabs(t_minus);
120                     n = normalize(2.0f * (p - c)); // Surface normal
121                 }
122             } // End of solution branch
123
124             // End of particle loop
125
126             // Write pixel color
127             if (tdist < 1e10) {
128                 // Lambertian shading parameters
129                 float dotprod = fabs(dot(n, constc_light));
130                 float I_d = 40.0f; // Light intensity
131                 float k_d = 5.0f; // Diffuse coefficient
132
133                 // Ambient shading
134                 float k_a = 10.0f;
135                 float I_a = 5.0f;
136
137                 // Write shading model values to pixel color ←
                    channels
138                 _img[mempos*4] = (unsigned char) ((k_d * ←
                    I_d * dotprod
139                 + k_a * I_a)*0.48f);
140                 _img[mempos*4 + 1] = (unsigned char) ((k_d * ←
                    I_d * dotprod
141                 + k_a * I_a)*0.41f);
142                 _img[mempos*4 + 2] = (unsigned char) ((k_d * ←
                    I_d * dotprod
143                 + k_a * I_a)*0.27f);
144             }
145         }
146     }
147 }
148
149 void cameraInit_cpu(float3 eye, float3 lookat, float ←
    imgw, float hwratio)
150 {
151     // Image dimensions in world space (l, r, b, t)
152     float4 imgplane = make_float4(-0.5f*imgw, ←
        0.5f*imgw, -0.5f*imgw*hwratio, ←
        0.5f*imgw*hwratio);

```



```

158 // The view vector
159 float3 view = eye - lookat;
160
161 // Construct the camera view orthonormal base
162 float3 v = make_float3(0.0f, 1.0f, 0.0f); // v: ←
163 // Pointing upward
164 float3 w = -view/lengthf3(view); // w: ←
165 // Pointing backwards
166 float3 u = cross(make_float3(v.x, v.y, v.z), ←
167 // make_float3(w.x, w.y, w.z)); // u: Pointing right
168
169 // Focal length 20% of eye vector length
170 float d = lengthf3(view)*0.8f;
171
172 // Light direction (points towards light source)
173 float3 light = ←
174 // normalize(-1.0f*eye*make_float3(1.0f, 0.2f, ←
175 // 0.6f));
176
177 std::cout << "Transferring camera values to ←
178 // constant memory\n";
179
180 constc_u = u;
181 constc_v = v;
182 constc_w = w;
183 constc_eye = eye;
184 constc_imgplane = imgplane;
185 constc_d = d;
186 constc_light = light;
187 }
188
189 // Wrapper for the rt algorithm
190 int rt_cpu(float4* p, unsigned int np,
191 // rgb* img, unsigned int width, unsigned int ←
192 // height,
193 // f3 origo, f3 L, f3 eye, f3 lookat, float imgw) {
194
195 using std::cout;
196
197 cout << "Initializing CPU raytracer:\n";
198
199 // Initialize GPU timestamp recorders
200 float t1_go, t2_go, t1_stop, t2_stop;
201
202 // Start timer 1
203 t1_go = clock();
204
205 // Allocate memory
206 cout << "Allocating device memory\n";
207 static unsigned char *_img; // RGBw values in ←
208 // image
209 static float3* _ray_origo; // Ray origo (x,y,z)
210 static float3* _ray_direction; // Ray direction ←
211 // (x,y,z)
212 _img = new unsigned char[width*height*4];
213 _ray_origo = new float3[width*height];
214 _ray_direction = new float3[width*height];
215
216 // Arrange thread/block structure
217 unsigned int pixels = width*height;
218 float hw_ratio = (float)height/(float)width;
219
220 // Start timer 2
221 t2_go = clock();
222
223 // Initialize image to background color
224 imageInit_cpu(_img, pixels);
225
226 // Initialize camera
227 cameraInit_cpu(make_float3(eye.x, eye.y, eye.z),
228 // make_float3(lookat.x, lookat.y, ←
229 // lookat.z),
230 // imgw, hw_ratio);
231
232 // Construct rays for perspective projection
233 rayInitPerspective_cpu(
234 // _ray_origo, _ray_direction,
235 // make_float3(eye.x, eye.y, eye.z),
236 // width, height);
237
238 // Find closest intersection between rays and spheres
239 rayIntersectSpheres_cpu(
240 // _ray_origo, _ray_direction,
241 // p, _img, pixels, np);
242
243 // Stop timer 2
244 t2_stop = clock();
245
246 memcpy(img, _img, sizeof(unsigned char)*pixels*4);
247
248 // Free dynamically allocated device memory
249 delete [] _img;
250 delete [] _ray_origo;
251 delete [] _ray_direction;
252
253 // Stop timer 1
254 t1_stop = clock();
255
256 // Report time spent
257 cout << "Time spent on entire CPU raytracing ←
258 // routine:\n"
259 // << (t1_stop-t1_go)/CLOCKS_PER_SEC*1000.0 << "←
260 // ms\n";
261 cout << "Functions:\n" << ←
262 // (t2_stop-t2_go)/CLOCKS_PER_SEC*1000.0 << "←
263 // ms\n";
264
265 // Return successfully
266 return 0;
267 }

```

APPENDIX C BLOOPERS

This section contains pictures of the unfinished raytracer in a malfunctioning state, which can provide interesting, however unusable, results.




Fig. 5. An end result encountered way too many times, e.g. after inadvertently dividing the focal length with zero.




Fig. 8. Colorful result caused by forgetting to normalize the normal vector calculation.




Fig. 6. Twisted appearance caused by wrong thread/block layout.




Fig. 7. A problem with the distance determination caused the particles to be displayed in a wrong order.




Fig. 9. Another colorful result with wrong order of particles, cause by using `abs()` (meant for integers) instead of `fabs()` (meant for floats).